Dialog zur Farbauswahl anzeigen
Versionen: Access 97, 2000, 2002/XP und 2003

 FRAGE:  Wir benötigen für verschiedene Aufgaben einen Dialog, über den die Anwender unserer Datenbank Farben auswählen können. Dabei sollte es möglich sein, benutzerdefinierte Farben zu verwenden und die aktuell gewählte Farbe als Vorgabe zu setzen.
H. Owens

 ANTWORT:  Gehen Sie wie folgt vor:

1. Legen Sie ein neues Modul "modColorDialog" an.

2. Geben Sie im allgemeinen Teil die folgenden Deklarationen ein:

Private Const CC_SOLIDCOLOR = &H80
Private Const CC_RGBINIT = &H1
Private Const CC_FULLOPEN = &H2

Type tCHOOSECOLOR
  lStructSize As Long
  hwnd As Long
  hInstance As Long
  rgbResult As Long
  lpCustColors As String
  Flags As Long
  lCustData As Long
  lpfnHook As Long
  lpTemplateName As String
End Type

Declare Function ChooseColor Lib "comdlg32.dll" _
                 Alias "ChooseColorA" _
                (pChoosecolor As tCHOOSECOLOR) As Long
· Geben Sie anschließend die folgende Funktion ein:

Function ColorDialog(Optional varCurrColor As Variant) _
                     As Long
  Dim CS As tCHOOSECOLOR
  Dim strCustColors As String * 64
  Dim R As Variant
  Dim lngCurrColor As Long
  Dim intRed As Integer, _
      intGreen As Integer, _
      intBlue As Integer

  ColorDialog = -1 'Default: Keine Auswahl
  strCustColors = String$(64, 0)
  With CS
    .lStructSize = Len(CS)
    .hwnd = Application.hWndAccessApp
    If IsMissing(varCurrColor) Then
      .Flags = CC_SOLIDCOLOR + _
               CC_FULLOPEN
    Else
      lngCurrColor = varCurrColor
      intRed = lngCurrColor Mod &H100
      lngCurrColor = lngCurrColor \ &H100
      intGreen = lngCurrColor Mod &H100
      lngCurrColor = lngCurrColor \ &H100
      intBlue = lngCurrColor Mod &H100
      Mid$(strCustColors, 1, 1) = Chr$(intRed)
      Mid$(strCustColors, 2, 1) = Chr$(intGreen)
      Mid$(strCustColors, 3, 1) = Chr$(intBlue)
      .Flags = CC_SOLIDCOLOR + _
               CC_RGBINIT + _
               CC_FULLOPEN
      .rgbResult = varCurrColor
    End If
    .lpCustColors = strCustColors
    R = ChooseColor(CS)
    If R <> 0 Then
      ColorDialog = .rgbResult
    End If
  End With

End Function

Um eine einfache Farbauswahl anzuzeigen, rufen Sie die Funktion "ColorDialog()" einfach ohne Parameter auf:

lngColor = ColorDialog()

Soll eine System- oder benutzerdefinierte Farbe voreingestellt werden, geben Sie die aktuelle Einstellung als Parameter an:

  lngColor = ColorDialog(10987431)
  lngColor = ColorDialog(Me.Feld.ForeColor)
  lngColor = ColorDialog(RGB(0, 255, 0))


Das Funktionsergebnis ist entweder die gewählte Farbe oder "-1", wenn auf Abbrechen geklickt wurde.
